unittest
{
	assert( [ "1000 8", "1 3 4 5 6 7 8 9" ].parse.expand.solve == 2000 );
	assert( [ "9999 1", "0" ].parse.expand.solve == 9999 );
	
	assert( [ "1112 1", "0 2 3 4 5 6 7 8 9" ].parse.expand.solve == 11111 );		// 追加ケース
}



import std.algorithm;
import std.conv;
import std.range;
import std.stdio;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	auto nk = input.front.split.to!( long[] );
	input.popFront;
	auto ds = input.front.split.to!( long[] );
	return tuple( nk[ 0 ], ds );
}

auto solve( long n, long[] ds )
{
	auto cs = n.to!string.map!( ( a ) => a - '0' ).retro.array;
	
	for( auto i = 0L; i < cs.length; i++ )
	{
		// 禁止数字判定
		foreach( d; ds )
		{
			if( d < cs[ i ] ) continue;
			if( cs[ i ] < d ) break;
			cs[ i ]++;
		}
		
		// 繰り上がり
		if( 10 <= cs[ i ] )
		{
			if( i + 1 < cs.length ) cs[ i + 1 ]++;
			else                    cs ~= 1L;
			cs[ i ] = 0L;
			i--;		// 今の桁を判定をやり直し
		}
	}
	
	return reduce!( ( a, b ) => a * 10 + b )( 0L, cs.retro );
}
